

Задача 7

(Задача 8.54. Сборник задач по теоретической механике/ Под ред. К.С. Колесникова. М.: Наука, Главная редакция физико-математической литературы, 1983. – 320 с.)


Тело массы m начинает движение в воде на глубине h с вертикальной скоростью v_0 и движется вверх, преодолевая силу \mathbf{R}_1 сопротивления воды. Модуль этой силы $R_1 = \mu_1 v^2$, где v – скорость тела, $\mu_1 = \text{const} > 0$. После выхода тела из воды на него в течении T секунд действует сила $\mathbf{F} = \text{const}$, линия действия которой в течении всего отрезка времени наклонена под углом $\alpha = 30^\circ$ к горизонту. Сила сопротивления воздуха силы $\mathbf{R}_2 = -\mu_2 \mathbf{v}$, где $\mu_2 = \text{const} > 0$.

Пренебрегая выталкивающей силой, действующей на тело в воде, найти высоту H его подъема над поверхностью воды в момент окончания действия силы \mathbf{F} , если $F = 4mg$,

$\mu_2 = 0.5\mu_1 v_0$, $h = \frac{m}{2\mu_1} \ln \frac{\mu_1 v_0^2}{mg}$ и поле сил тяжести однородно.

Для решения задачи использовать следующие значения параметров:

$m = 10 \text{ кг}$, $v_0 = 100 \text{ м/с}$, $c = 10 \text{ Н/м}$, $\alpha = 30 \text{ град}$, $\mu_1 = 1 \text{ Нс}^2 / \text{м}^2$, $T = 0.5 \text{ с}$.



Точный теоретический ответ

Высота определяется по формуле

$$H = \frac{m^2 \cdot g}{2 \cdot \mu_2^2} \cdot \left(e^{-\frac{\mu_2 \cdot T}{m}} - 1 \right) + \frac{m \cdot g}{\mu_2} \cdot T$$

Решение задачи в EULER



Система состоит из двух звеньев.

- Инерциальное звено (body1). В проекте инерциальное звено не отображается.
- Тело (body2). Тело изображается точкой point1 и имеет массу m .

В проекте задана гравитация с ускорением свободного падения $9.81 \text{ [м/с}^2\text{]}$.

Начальная скорость телу сообщается с помощью объекта «условие состояния механизма» condition1. Сила сопротивления воды моделируется силовым элементом force1 типа «сила

по датчикам на одну точку» с функцией силы function1. Сила, толкающая тело, моделируется силовым элементом force2 типа «сила по датчикам на одну точку». Сила сопротивления воздуха моделируется силовыми элементами force3 и force4 типа «сила по датчикам на одну точку» с функцией силы function2. Момент выхода тела из воды моделируется событием event1. Оно выключает силовой элемент force1 и включает силовые элементы force2, force3, force4 и датчик относительного времени event1.timer. Для определения высоты тела используется датчик H. Для остановки расчета используется событие event2.

Для решения задачи используется команда command1 «расчет динамики движения» с временем интегрирования 2 [s] и заданным постоянным шагом.

Результаты моделирования

Относительное отличие решения задачи в EULER в зависимости от шага численного интегрирования (использовался постоянный шаг интегрирования) представлено в следующей таблице.

Шаг интегрирования [s]	Относительное отличие от теоретического решения H_delta_rel [-]
0.01	0.000 022
0.001	0.000 000 84
0.000 1	0.000 005 5
0.000 01	0.000 003 9
0.000 001	0.000 000 90

Текст проекта в EULER

```

scalar m=10 [ kg ];
scalar v0=100 [ m/ s ];
scalar T=0.5 [ s ];
scalar a=30 [ deg ];
scalar mu1=1 [ N s2/ m2 ];
scalar mu2=0.5*mu1*v0;
scalar g=9.81 [ m/ s2 ];
scalar h=0.5*m*ln( mu1*v0*v0/ m/ g )/ mu1;
scalar F=4*m*g;
body body1=body( color = RGB( 229, 229, 229 ) );
set ground = body1;
point point1=point( 0 [ m ], 0 [ m ], 0 [ m ] );
solid solid1=sphere( point1, 0.1 [ m ], mass = m );
body body2=body( color = RGB( 255, 153, 153 ) );
body body2 < ( point1, solid1 );
point point2=point( 0 [ m ], h, 0 [ m ] );
gravity gravity1=parallel( reverse( projectY ), g = g );
condition condition1=transVelocity( body1, projectY, body2, point1, v0 );
function function1(v [ m/ s ])= -mu1*#v*#v;
sensor vx=transVelocity( body1, projectX, body2, point1 );
sensor vy=transVelocity( body1, projectY, body2, point1 );
force force1=force( body2, point1, projectY, function1, list( vy ) );
vector vector1=rotate( projectX, projectZ, a );
function function2(v [ m/ s ])= -mu2*#v;
force force2=force( body2, point1, vector1, F, list( ), work = off: );
force force3=force( body2, point1, projectX, function2, list( vx ), work = off: );
force force4=force( body2, point1, projectY, function2, list( vy ), work = off: );
sensor H=bodyDisplacement( body1, point2, projectY, body2, point1 );
event event1=reformsBySensor( list( complex( list( forcesOff( list( force1 ) ), forcesOn( list( force2, force3, force4 ) ) ) ) ), H,
0 [ m ], timer = timerOn: );
event event2=reformsBySensor( list( stop( ) ), event1.timer, T );
sensor H_theoretical=0.5*m*m*g/ mu2/ mu2*( exp( -mu2*T/ m ) - 1 )+m*g*T/ mu2;
sensor H_delta_rel=abs( ( H -H_theoretical )/ H_theoretical );
command command1=constRK4( 2.00000e+000 [ s ], 1.00000e-003 [ s ] );

```

Λ///
Λ Список главных команд;
set dynamics = command1;

Λ///
Λ Единицы измерения;
set units = SI;